technical memorandum Daresbury Laboratory

DL/SCI/TM54A

PMU3D - A PROGRAM FOR THREE-DIMENSIONAL FIELD CALCULATIONS ON PERIODIC PERMANENT MAGNET SYSTEMS. (DL/SCI/TM29A UPDATE)

by

R.P. WALKER, SERC Daresbury Laboratory

NOVEMBER, 1987

Science & Engineering Research Council

Daresbury Laboratory

Daresbury, Warrington WA4 4AD

ISSN 0144-5677

1. INTRODUCTION

In the last few years many undulator magnets have been constructed at various laboratories both to provide sources of synchrotron radiation in electron storage rings and also for use in free electron lasers. The majority of these devices have employed permanent magnet rather than electromagnet technology, based largely on the use of rare earth cobalt (REC) material which has a higher remanent field compared to earlier types of permanent magnet material. Two different configurations have been developed, the first using only permanent magnet material, "pure REC" design, and the second using iron poles, "hybrid" design. In the former case it is possible to compute the field distribution more simply than if iron (or non-linear permanent magnet material) were involved because of the particular properties of REC. A program for computing the 3-dimensional magnetic field distribution in undulator magnets of this type (PMU3D) was described in ref.1 and since then has been used extensively at Daresbury Laboratory (with some modifications) in the design of the undulator magnets for the SRS[2] and the UK FEL project[3].

It is the purpose of this report to provide an up-to-date user guide for the program together with listings and descriptions of the relevant routines to enable the user to adapt the program to his own needs. An example is given of a program based on the PMU3D routines that has been of use in the design of practical undulator magnets.

The program has been developed on the IBM370 and AS/7000 central computer and was originally written in FORTRAN IV using various IBM extensions. It has now been converted to standard FORTRAN 77 for compatibiiity with other computer systems.

2. MATHEMATICAL BASIS OF PMU3D

The model used for the permanent magnet material[4] assumes that:

i) The permeability of the materiai is unity, so that it may be treated iike vacuum with an imprinted current density. This means that the field produced by different pieces superpose linearly.

ii) The blocks are homogeneously magnetised, so that they may be represented by current sheets at the block surfaces only.

Both of these are very good approximations for REC blocks.

The current sheet model gives the magnetic field produced by a single piece of REC as the following surface integral:

$$\underline{B} = \frac{1}{4\pi} \int \frac{(\underline{M} \wedge \underline{dA}) \wedge \underline{r}}{r^3}$$

where $\underline{M}$ is the magnetisation (strictly μ_o times the magnetisation)
$\underline{dA}$ is a surface area element, normal to the surface and directed out of the material
$\underline{r}$ is the vector from a point on the surface to the field point

For parallelepiped blocks with rectangular faces lying parallel to the axes of the coordinate system the integration can be performed analytically (a derivation is given in ref.1) resulting in the following expression for the magnetic field components:

$$B_i = \frac{-1}{4\pi} \sum_{\substack{j=1 \\ k\neq i\neq j}}^{3} M_i \left[\tan^{-1}\left(\frac{r_i r_k}{r_j r}\right) \right]_{r_{i1}r_{j1}r_{k1}}^{r_{i2}r_{j2}r_{k2}} + M_j \left[\ln(r_k+r) \right]_{r_{i1}r_{j1}r_{k1}}^{r_{i2}r_{j2}r_{k2}} \qquad (1)$$

In the above, i, j and k is some permutation of integers 1, 2, 3 representing the x, y, z components respectively. $\underline{r}$ is a vector from the block surface to the field point. The expressions in brackets are evaluated at the block corners, defined by components r_{i1}, r_{j1}, r_{k1} and r_{i2}, r_{j2}, r_{k2}.

It is clear from the equation above that problems can arise in computing the magnetic field under various circumstances:

i) when $r_j = 0$ in the first term, i.e. points lying in the plane containing a block surface,
ii) when $r_k = -r$ in the second term, i.e. points lying on lines extending from the block edges, or
iii) when $r = 0$ in either term, i.e. at any of the block corners.

The first situation is handled in the program by setting the contribution of the relevant block face to zero, but the other conditions cannot be handled so easily. However, in these situations it is only necessary to offset the field point by a very small amount ($\lesssim 10^{-3}$ mm) to eliminate any problems.

3. MAIN PROGRAM DESCRIPTION

The main program PMU3D to calculate each of the magnetic field components over a specified range of coordinate values, as well as field integrals, is given in Appendix 1. The input parameters are as follows, read using free-format:

PERIOD	GAP	HEIGHT	WIDTH		SPACE		
BR	M	NP					
XMIN	DX	NX					
YMIN	DY	NY					
ZMIN	DZ	NZ					
BX	BY	BZ	I1	I2	I3	OPT	

The parameters on the first line define the magnet geometry, as illustrated in fig.1 which also shows the coordinate system with its origin at the centre of the magnet. On the following line BR is the remanent field of the REC material (in tesla). M is the number of blocks in each array per period, i.e. the magnetization angle rotates by $2\pi/M$ per block. In the usual arrangement therefore (fig.1) M = 4. NP is the number of periods in each array. It is assumed that the blocks at either end of the magnet are half the length of the standard blocks, as shown, and have the same magnetisation strength which guarantees zero field integral, $\int B_y dz = 0$, in the idealised magnet. The following 3 lines define the region over which the field is to be calculated; for each coordinate in turn the minimum value, the step size and the number of points are given. The magnet dimensions and the coordinate values XMIN, DX etc. may be given in any unit, provided the same unit is used for all of the physical parameters.

On the final line 7 integers are required. A non zero value of BX, BY or BZ indicates that the corresponding field component is to be calculated.

I1, I2 and I3 is some permutation of the integers 1, 2 and 3, corresponding to coordinate axes x, y and z respectively, which defines the formatting of the calculated field values. The data are presented in a series of tables, one for each field component and for each value of coordinate I3, with coordinate I1 varying down the page and coordinate I2 varying across the page. Such a definition allows the user great flexibility for structuring the output. If OPT = 1 an integration of each field component will be made along the I1 coordinate direction, for each value of coordinates I2 and I3, using the trapezoidal rule. The accuracy of the result will of course depend on the step size chosen for the I1 coordinate and the user should vary this to check that the required accuracy has been obtained.

Although not strictly necessary the program stores the I1 coordinate values in C1 and the field values in the 2-dimensional array B, for use in other routines that the user may wish to include, e.g. Fourier analysis.

As the program is presently defined the total number of blocks per array is limited to 401, i.e.

$$(NP \times M) + 1 < 401$$

which allows up to 100 periods for an arrangement with M = 4. The maximum number of field values that can be calculated in the I1 direction is 200. There is no limit on the number of values in the I2 direction, however only 10 values will fit on each page.

The program occupies 152 K of storage on the AS/7000 and for a general field point takes on average approximately 0.6 msec of CPU time to calculate each field component for every REC block in the undulator structure, including both upper and lower arrays of blocks.

Appendix 2 gives an example of a set of input parameters for calculating the field distribution in the central period of a typical undulator together with the corresponding output. Physical dimensions are given in mm, and the integrated field values are therefore in T mm.

4. DESCRIPTIONS OF SUBROUTINES

PMU3D calls SETPMU (Appendix 3) to set up the coordinates and magnetization components for each individual block in the top array of blocks and store the data in COMMON/PMUDAT/. The coordinates are defined with respect to the origin at the centre of the magnet (see fig.1). For the calculation of an individual field component at a given set of coordinates (C) the subroutine PMU is called (Appendix 4). PMU sets up the coordinates for each block in turn in vectors V1 and V2 (see fig.2) and the magnetisation in M and calls PMB (Appendix 5) to calculate the field due to each individual block. Median plane symmetry is imposed i.e. when y = 0, B_x and B_z are set to zero and to save computational time B_y is calculated from the top array of blocks only, the result then being multiplied by a factor of 2.

PMB is the central part of PMU3D and is a general purpose routine to calculate the field due to a single REC block. It may be used independently of the preceding routines, for example to enable a comparison to be made between the predicted and measured field at specific points which may then be used to deduce the magnetisation components of the block[2]. PMB evaluates the expression given earlier for the magnetic field, eq.(1). It first sets up relative vectors R1 and R2 from the block corners to the field point (see fig.2). For efficiency components R1(I) etc. are copied into a set of variables R1I etc. and the distance from each block corner to the field point (A1-A8) are calculated, the data being stored in COMMON/PMBDAT/. Functions I1 and I2 calculate the first and second terms in eq.(1) respectively. The limits of the expressions are evaluated as follows:

$$[F(x,y,z)]_{x_1y_1z_1}^{x_2y_2z_2} = F(x_2y_2z_2) + F(x_1y_1z_2) + F(x_1y_2z_1) + F(x_2y_1z_1)$$

$$- [F(x_1y_2z_2) + F(x_2y_1z_2) + F(x_2y_2z_1) + F(x_1y_1z_1)]$$

5. UNDULATOR DESIGN PROGRAM

In the process of designing an undulator magnet it is often necessary to survey a wide range of different magnet geometries and examine particular characteristics of the field distribution. Rather than calculate the distributions for each magnet geometry separately using PMU3D a better solution is to write a specific program to do the required calculations using the basic subroutines described earlier. An example of such a program is given in Appendix 6. In this case the user can vary any of the magnet period, gap, height, width and block spacing parameters and calculate the field amplitude on axis (B0), K value and field homogeneity in the x and y directions. The program includes the option that if a zero block height is specified a square cross-section will be assumed, the height being calculated from the specified PERIOD and SPACE parameters. The field homogeneity is defined as follows:

$$a_x = -\frac{1}{2}\frac{d^2B_y}{dx^2} \qquad a_y = \frac{1}{2}\frac{d^2B_y}{dy^2}$$

Since it is known that the field varies quadratically with transverse position close to the z axis it is only necessary to evaluate the field at three points, namely the origin and points displaced by a specified small distance (R) in the x and y directions.

The program also calculates the field amplitude in the limit of infinite block width (B2D), together with the corresponding K value, using the following analytic expression[5]:

$$B_0 = 2B_r \frac{\mathrm{Sin}\ \varepsilon\ \pi/M}{\pi/M}\left(1 - e^{-\frac{2\pi h}{\lambda_0}}\right) e^{-\frac{\pi g}{\lambda_0}}$$

where λ_0, g, h are the PERIOD, GAP, and HEIGHT respectively while ε is related to the SPACE parameter (Δ):

$$\varepsilon = 1 - \frac{M\Delta}{\lambda_0}$$

A sample output from the program is given in Appendix 7. In this example a square cross-section block has been assumed. Dimensions are in mm, and the unit of a_x and a_y is % mm^{-2}.

REFERENCES

1. M.W. Fan, M.W. Poole and R.P. Walker, Daresbury Int. Rep., DL/SCI/TM29A.
2. R.P. Walker, M.W. Poole, and D.G. Taylor, J. Phys. (Paris), Coll. C1, 45 (1984) 321.
3. M.W. Poole, R.J. Bennett and R.P. Walker, J. Phys. (Paris), Col. C1, 45 (1984) 325.
4. K. Halbach, Nucl. Instr. Meth. 169 (1980) 1.
5. K. Halbach, Nucl. Instrum. Meth. 187 (1981) 109.

FIGURE CAPTIONS

Fig. 1 Undulator magnet geometry and coordinate system

Fig. 2 Definition of coordinates for an individual permanent magnet block

Appendix 1. Main PMU3D Program.

```
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      INTEGER CODE(3),OPT
      CHARACTER L(3)
      DIMENSION CMIN(3),DC(3),C(3),NC(3),C1(200),C2(10),AREA(10),
     $ B(200,10)
      DATA L/'X','Y','Z'/
C
C     READ INPUT PARAMETERS
C
      READ(5,*)PERIOD,GAP,HEIGHT,WIDTH,SPACE
      WRITE(6,1000)PERIOD,GAP,HEIGHT,WIDTH,SPACE
      READ(5,*)BR,M,NP
      WRITE(6,1100)BR,M,NP
      DO 10 I=1,3
      READ(5,*)CMIN(I),DC(I),NC(I)
   10 CONTINUE
      WRITE(6,1200)(CMIN(I),DC(I),NC(I),I=1,3)
      READ(5,*)CODE,I1,I2,I3,OPT
      WRITE(6,1300)CODE,I1,I2,I3,OPT
C
      N1=NC(I1)
      N2=NC(I2)
      N3=NC(I3)
      NPAGE=((N2-1)/10)+1
C
C     SET UP UNDULATOR GEOMETRY
C
      CALL SETPMU(PERIOD,GAP,HEIGHT,WIDTH,SPACE,BR,M,NP)
C
C     VARY FIELD COMPONENT
C
      DO 90 I=1,3
      IF(CODE(I).EQ.0)GOTO 90
C
C     VARY I3 COORDINATE
C
      DO 80 IC=1,N3
      C(I3)=CMIN(I3)+((IC-1)*DC(I3))
C
C     SET PAGE NO. AND I2 COORDINATES
C
      DO 70 N=1,NPAGE
      IF(N.NE.NPAGE)N22=10
      IF(N.EQ.NPAGE)N22=N2-((NPAGE-1)*10)
      DO 20 IB=1,N22
      C2(IB)=CMIN(I2)+((((N-1)*10)+IB-1)*DC(I2))
   20 CONTINUE
      IF(NPAGE.EQ.1)WRITE(6,2000)L(I),L(I3),C(I3)
      IF(NPAGE.GT.1)WRITE(6,2100)L(I),L(I3),C(I3),N
      WRITE(6,2200)L(I1),L(I2),(C2(IB),IB=1,N22)
C
C     VARY I1 AND I2 COORDINATES
C     AND DO THE FIELD CALCULATION
C
      DO 40 IA=1,N1
      C(I1)=CMIN(I1)+((IA-1)*DC(I1))
      C1(IA)=C(I1)
      DO 30 IB=1,N22
      C(I2)=C2(IB)
      CALL PMU(C,B(IA,IB),I)
   30 CONTINUE
      WRITE(6,3000)C(I1),(B(IA,IB),IB=1,N22)
   40 CONTINUE
C
C     CARRY OUT INTEGRATION IF REQUIRED
C
      IF(OPT.NE.1)GOTO 70
      DO 60 IB=1,N22
      AREA(IB)=-(B(1,IB)+B(N1,IB))/2.0
      DO 50 IA=1,N1
      AREA(IB)=AREA(IB)+B(IA,IB)
   50 CONTINUE
      AREA(IB)=AREA(IB)*DC(I1)
   60 CONTINUE
      WRITE(6,4000)(AREA(IB),IB=1,N22)
   70 CONTINUE
   80 CONTINUE
   90 CONTINUE
      STOP
 1000 FORMAT('1******   PMU3D   ******'//1X,
     $ 'PERIOD = ',F7.2,2X,'GAP = ',F7.2,2X,'HEIGHT = ',F7.2,
     $ 2X,'WIDTH = ',F7.2,2X,'SPACE = ',F7.2)
 1100 FORMAT(/1X,'BR = ',F6.4,2X,'M = ',I2,2X,'NP = ',I3)
 1200 FORMAT(/1X,'AREA FOR CALCULATION :'/
     $ 1X,'X : ',2(F7.2,2X),I3/
     $ 1X,'Y : ',2(F7.2,2X),I3/
     $ 1X,'Z : ',2(F7.2,2X),I3)
 1300 FORMAT(/1X,'FIELD COMPONENTS : ',3(I1,1X),2X,'OUTPUT FORMAT : ',
     $ 3(I1,1X),2X,'OPT : ',I1)
 2000 FORMAT(//1X,A1,' COMPONENT OF MAGNETIC FIELD AT ',A1,' = ',F5.1)
 2100 FORMAT(//1X,A1,' COMPONENT OF MAGNETIC FIELD AT ',A1,' = ',F5.1,
     $ ' - PAGE ',I1)
 2200 FORMAT(/5X,A1,3X,A1,':',2X,10(F6.2,5X))
 3000 FORMAT(1X,F7.2,2X,10(1X,D10.3))
 4000 FORMAT(/1X,'AREA',5X,10(1X,D10.3))
      END
```

Appendix 2. Sample input parameters for PMU3D.

```
100.0 50.0 25.0 50.0 0.0
1.0 4 10
-20.0 20.0  3
-10.0 10.0 3
-50.0 10.0  11
1 1 1 3 1 2 1
```

Sample output :

```
******   PMU3D   ******

PERIOD =  100.00  GAP =   50.00  HEIGHT =   25.00  WIDTH =   50.00  SPACE =    0.00

BR = 1.0000  M =  4  NP = 10

AREA FOR CALCULATION :
X :  -20.00     20.00     3
Y :  -10.00     10.00     3
Z :  -50.00     10.00    11

FIELD COMPONENTS : 1 1 1   OUTPUT FORMAT : 3 1 2   OPT : 1


X COMPONENT OF MAGNETIC FIELD AT Y = -10.0

     Z   X:  -20.00        0.00       20.00
  -50.00     0.695D-01  0.000D+00 -0.695D-01
  -40.00     0.564D-01  0.000D+00 -0.564D-01
  -30.00     0.212D-01  0.000D+00 -0.212D-01
  -20.00    -0.212D-01  0.000D+00  0.212D-01
  -10.00    -0.564D-01  0.000D+00  0.564D-01
    0.00    -0.695D-01  0.000D+00  0.695D-01
   10.00    -0.564D-01  0.000D+00  0.564D-01
   20.00    -0.212D-01  0.000D+00  0.212D-01
   30.00     0.212D-01  0.000D+00 -0.212D-01
   40.00     0.564D-01  0.000D+00 -0.564D-01
   50.00     0.695D-01  0.000D+00 -0.695D-01

AREA         0.252D-05  0.000D+00 -0.252D-05


X COMPONENT OF MAGNETIC FIELD AT Y =   0.0

     Z   X:  -20.00        0.00       20.00
  -50.00     0.000D+00  0.000D+00  0.000D+00
  -40.00     0.000D+00  0.000D+00  0.000D+00
  -30.00     0.000D+00  0.000D+00  0.000D+00
  -20.00     0.000D+00  0.000D+00  0.000D+00
  -10.00     0.000D+00  0.000D+00  0.000D+00
    0.00     0.000D+00  0.000D+00  0.000D+00
   10.00     0.000D+00  0.000D+00  0.000D+00
   20.00     0.000D+00  0.000D+00  0.000D+00
   30.00     0.000D+00  0.000D+00  0.000D+00
   40.00     0.000D+00  0.000D+00  0.000D+00
   50.00     0.000D+00  0.000D+00  0.000D+00

AREA         0.000D+00  0.000D+00  0.000D+00
```



```
X COMPONENT OF MAGNETIC FIELD AT Y =  10.0

     Z   X:  -20.00        0.00       20.00
  -50.00    -0.695D-01  0.000D+00  0.695D-01
  -40.00    -0.564D-01  0.000D+00  0.564D-01
  -30.00    -0.212D-01  0.000D+00  0.212D-01
  -20.00     0.212D-01  0.000D+00 -0.212D-01
  -10.00     0.564D-01  0.000D+00 -0.564D-01
    0.00     0.695D-01  0.000D+00 -0.695D-01
   10.00     0.564D-01  0.000D+00 -0.564D-01
   20.00     0.212D-01  0.000D+00 -0.212D-01
   30.00    -0.212D-01  0.000D+00  0.212D-01
   40.00    -0.564D-01  0.000D+00  0.564D-01
   50.00    -0.695D-01  0.000D+00  0.695D-01

AREA        -0.252D-05  0.000D+00  0.252D-05


Y COMPONENT OF MAGNETIC FIELD AT Y = -10.0

     Z   X:  -20.00        0.00       20.00
  -50.00    -0.223D+00 -0.305D+00 -0.223D+00
  -40.00    -0.182D+00 -0.249D+00 -0.182D+00
  -30.00    -0.676D-01 -0.929D-01 -0.676D-01
  -20.00     0.676D-01  0.929D-01  0.676D-01
  -10.00     0.182D+00  0.249D+00  0.182D+00
    0.00     0.223D+00  0.305D+00  0.223D+00
   10.00     0.182D+00  0.249D+00  0.182D+00
   20.00     0.676D-01  0.929D-01  0.676D-01
   30.00    -0.676D-01 -0.929D-01 -0.676D-01
   40.00    -0.182D+00 -0.249D+00 -0.182D+00
   50.00    -0.223D+00 -0.305D+00 -0.223D+00

AREA        -0.455D-03 -0.458D-03 -0.455D-03


Y COMPONENT OF MAGNETIC FIELD AT Y =   0.0

     Z   X:  -20.00        0.00       20.00
  -50.00    -0.176D+00 -0.240D+00 -0.176D+00
  -40.00    -0.143D+00 -0.194D+00 -0.143D+00
  -30.00    -0.544D-01 -0.740D-01 -0.544D-01
  -20.00     0.543D-01  0.740D-01  0.543D-01
  -10.00     0.143D+00  0.194D+00  0.143D+00
    0.00     0.176D+00  0.240D+00  0.176D+00
   10.00     0.143D+00  0.194D+00  0.143D+00
   20.00     0.543D-01  0.740D-01  0.543D-01
   30.00    -0.544D-01 -0.740D-01 -0.544D-01
   40.00    -0.143D+00 -0.194D+00 -0.143D+00
   50.00    -0.176D+00 -0.240D+00 -0.176D+00

AREA        -0.456D-03 -0.458D-03 -0.456D-03


Y COMPONENT OF MAGNETIC FIELD AT Y =  10.0

     Z   X:  -20.00        0.00       20.00
  -50.00    -0.223D+00 -0.305D+00 -0.223D+00
  -40.00    -0.182D+00 -0.249D+00 -0.182D+00
```

-30.00	-0.676D-01	-0.929D-01	-0.676D-01
-20.00	0.676D-01	0.929D-01	0.676D-01
-10.00	0.182D+00	0.249D+00	0.182D+00
0.00	0.223D+00	0.305D+00	0.223D+00
10.00	0.182D+00	0.249D+00	0.182D+00
20.00	0.676D-01	0.929D-01	0.676D-01
30.00	-0.676D-01	-0.929D-01	-0.676D-01
40.00	-0.182D+00	-0.249D+00	-0.182D+00
50.00	-0.223D+00	-0.305D+00	-0.223D+00
AREA	-0.455D-03	-0.458D-03	-0.455D-03

Z COMPONENT OF MAGNETIC FIELD AT Y = -10.0

Z	X: -20.00	0.00	20.00
-50.00	-0.533D-06	-0.663D-06	-0.533D-06
-40.00	-0.710D-01	-0.968D-01	-0.710D-01
-30.00	-0.114D+00	-0.156D+00	-0.114D+00
-20.00	-0.114D+00	-0.156D+00	-0.114D+00
-10.00	-0.710D-01	-0.968D-01	-0.710D-01
0.00	0.388D-06	0.530D-06	0.388D-06
10.00	0.710D-01	0.968D-01	0.710D-01
20.00	0.114D+00	0.156D+00	0.114D+00
30.00	0.114D+00	0.156D+00	0.114D+00
40.00	0.710D-01	0.968D-01	0.710D-01
50.00	-0.243D-06	-0.398D-06	-0.243D-06
AREA	0.362D-10	0.302D-10	0.362D-10

Z COMPONENT OF MAGNETIC FIELD AT Y = 0.0

Z	X: -20.00	0.00	20.00
-50.00	0.000D+00	0.000D+00	0.000D+00
-40.00	0.000D+00	0.000D+00	0.000D+00
-30.00	0.000D+00	0.000D+00	0.000D+00
-20.00	0.000D+00	0.000D+00	0.000D+00
-10.00	0.000D+00	0.000D+00	0.000D+00
0.00	0.000D+00	0.000D+00	0.000D+00
10.00	0.000D+00	0.000D+00	0.000D+00
20.00	0.000D+00	0.000D+00	0.000D+00
30.00	0.000D+00	0.000D+00	0.000D+00
40.00	0.000D+00	0.000D+00	0.000D+00
50.00	0.000D+00	0.000D+00	0.000D+00
AREA	0.000D+00	0.000D+00	0.000D+00

Z COMPONENT OF MAGNETIC FIELD AT Y = 10.0

Z	X: -20.00	0.00	20.00
-50.00	0.533D-06	0.663D-06	0.533D-06
-40.00	0.710D-01	0.968D-01	0.710D-01
-30.00	0.114D+00	0.156D+00	0.114D+00
-20.00	0.114D+00	0.156D+00	0.114D+00
-10.00	0.710D-01	0.968D-01	0.710D-01
0.00	-0.388D-06	-0.530D-06	-0.388D-06
10.00	-0.710D-01	-0.968D-01	-0.710D-01
20.00	-0.114D+00	-0.156D+00	-0.114D+00

30.00	-0.114D+00	-0.156D+00	-0.114D+00
40.00	-0.710D-01	-0.968D-01	-0.710D-01
50.00	0.243D-06	0.398D-06	0.243D-06
AREA	-0.362D-10	-0.302D-10	-0.362D-10

Appendix 6. Sample undulator design program.

```
      IMPLICIT DOUBLE PRECISION (A-H,K,O-Z)
      DIMENSION PERIOD(20),SPACE(20),GAP(20),HEIGHT(20),WIDTH(20),C(3)
      DATA C/3*0.0/
C
C     READ DATA
C
      READ(5,*)BR,M,NP,R
      READ(5,*)NPRIOD,NSPACE,NHT,NGAP,NWIDTH
      READ(5,*)(PERIOD(I),I=1,NPRIOD)
      READ(5,*)(SPACE(I),I=1,NSPACE)
      READ(5,*)(HEIGHT(I),I=1,NHT)
      READ(5,*)(GAP(I),I=1,NGAP)
      READ(5,*)(WIDTH(I),I=1,NWIDTH)
      WRITE(6,1000)BR,M,NP,R
C
C     VARY PERIOD
C
      DO 10 IA=1,NPRIOD
C
C     VARY SPACE BETWEEN BLOCKS
C
      DO 10 IB=1,NSPACE
C
C     VARY BLOCK HEIGHT
C
      DO 10 IC=1,NHT
      HIGHT=HEIGHT(IC)
      IF(HIGHT.EQ.0.0)HIGHT=(PERIOD(IA)/DFLOAT(M))-SPACE(IB)
C
C     VARY GAP
C
      DO 10 ID=1,NGAP
C
C     CALCULATE 2D B0 AND K VALUE
C
      B2D=B(BR,PERIOD(IA),GAP(ID),HIGHT,SPACE(IB),M)
      K2D=0.0934*PERIOD(IA)*B2D
C
C     VARY WIDTH
C
      WRITE(6,2000)
      DO 10 IE=1,NWIDTH
C
C     SET UP GEOMETRY AND DO FIELD CALCULATIONS
C
      CALL SETPMU(PERIOD(IA),GAP(ID),HIGHT,WIDTH(IE),SPACE(IB),
     $ BR,M,NP)
      CALL PMU(C,B0,2)
      K=0.0934*PERIOD(IA)*B0
      C(1)=R
      CALL PMU(C,B1,2)
      C(1)=0.0
      C(2)=R
      CALL PMU(C,B2,2)
      C(2)=0.0
      AX=(B0-B1)*100.0/(B0*R*R)
      AY=(B2-B0)*100.0/(B0*R*R)
      WRITE(6,3000)PERIOD(IA),SPACE(IB),HIGHT,GAP(ID),WIDTH(IE),
     $ B2D,B0,K2D,K,AX,AY
   10 CONTINUE
      STOP
 1000 FORMAT('1BR = ',F6.4,2X,'M = ',I2,2X,'NP = ',I2,2X,'R = ',F5.1//
     $ 1X,'PERIOD',2X,'SPACE',2X,'HEIGHT',4X,'GAP',4X,'WIDTH',3X,
     $ 'B2D',6X,'B0',5X,'K2D',5X,'K',8X,'AX',6X,'AY'/)
 2000 FORMAT(1X)
 3000 FORMAT(1X,F6.2,2X,F5.2,2X,F6.2,2X,F6.2,2X,F6.2,2X,F6.2,6(2X,F6.4))
      END



      FUNCTION B(BR,PERIOD,GAP,HEIGHT,SPACE,M)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DATA PI/3.1415926536/
C
C     CALCULATE FIELD AMPLITUDE USING 2D ANALYTIC EQUATION
C
      E=1.0-((M*SPACE)/PERIOD)
      E1=DSIN(E*PI/M)/(PI/M)
      E2=1.0-DEXP(-(2.0*PI*HEIGHT/PERIOD))
      E3=DEXP(-PI*GAP/PERIOD)
      B=2.0*BR*E1*E2*E3
      RETURN
      END
```

Appendix 7. Sample output from undulator design program.

BR = 0.9500 M = 4 NP = 10 R = 5.0

PERIOD	SPACE	HEIGHT	GAP	WIDTH	B2D	B0	K2D	K	AX	AY
65.00	0.00	16.25	20.00	10.00	0.5154	0.2273	3.1289	1.3799	0.7357	1.3170
65.00	0.00	16.25	20.00	20.00	0.5154	0.3710	3.1289	2.2522	0.3807	0.7609
65.00	0.00	16.25	20.00	30.00	0.5154	0.4440	3.1289	2.6958	0.1707	0.5420
65.00	0.00	16.25	20.00	40.00	0.5154	0.4792	3.1289	2.9093	0.0763	0.4554
65.00	0.00	16.25	20.00	50.00	0.5154	0.4961	3.1289	3.0119	0.0350	0.4189
65.00	0.00	16.25	20.00	60.00	0.5154	0.5043	3.1289	3.0618	0.0165	0.4028
65.00	0.00	16.25	20.00	70.00	0.5154	0.5084	3.1289	3.0866	0.0080	0.3954
65.00	0.00	16.25	20.00	80.00	0.5154	0.5105	3.1289	3.0990	0.0039	0.3920
65.00	0.00	16.25	20.00	90.00	0.5154	0.5115	3.1289	3.1054	0.0020	0.3903
65.00	0.00	16.25	20.00	100.00	0.5154	0.5121	3.1289	3.1087	0.0010	0.3895
65.00	0.00	16.25	50.00	10.00	0.1209	0.0326	0.7340	0.1977	0.2705	0.7845
65.00	0.00	16.25	50.00	20.00	0.1209	0.0607	0.7340	0.3686	0.2206	0.7204
65.00	0.00	16.25	50.00	30.00	0.1209	0.0822	0.7340	0.4989	0.1612	0.6490
65.00	0.00	16.25	50.00	40.00	0.1209	0.0970	0.7340	0.5890	0.1085	0.5894
65.00	0.00	16.25	50.00	50.00	0.1209	0.1066	0.7340	0.6471	0.0690	0.5467
65.00	0.00	16.25	50.00	60.00	0.1209	0.1125	0.7340	0.6829	0.0422	0.5187
65.00	0.00	16.25	50.00	70.00	0.1209	0.1160	0.7340	0.7043	0.0251	0.5013
65.00	0.00	16.25	50.00	80.00	0.1209	0.1181	0.7340	0.7168	0.0147	0.4908
65.00	0.00	16.25	50.00	90.00	0.1209	0.1193	0.7340	0.7240	0.0085	0.4846
65.00	0.00	16.25	50.00	100.00	0.1209	0.1199	0.7340	0.7282	0.0049	0.4810
65.00	1.00	15.25	20.00	10.00	0.4768	0.2127	2.8949	1.2914	0.7461	1.3679
65.00	1.00	15.25	20.00	20.00	0.4768	0.3461	2.8949	2.1010	0.3832	0.7871
65.00	1.00	15.25	20.00	30.00	0.4768	0.4131	2.8949	2.5081	0.1701	0.5607
65.00	1.00	15.25	20.00	40.00	0.4768	0.4450	2.8949	2.7018	0.0752	0.4721
65.00	1.00	15.25	20.00	50.00	0.4768	0.4602	2.8949	2.7939	0.0341	0.4352
65.00	1.00	15.25	20.00	60.00	0.4768	0.4675	2.8949	2.8384	0.0159	0.4190
65.00	1.00	15.25	20.00	70.00	0.4768	0.4711	2.8949	2.8602	0.0076	0.4117
65.00	1.00	15.25	20.00	80.00	0.4768	0.4729	2.8949	2.8711	0.0037	0.4083
65.00	1.00	15.25	20.00	90.00	0.4768	0.4738	2.8949	2.8767	0.0019	0.4067
65.00	1.00	15.25	20.00	100.00	0.4768	0.4743	2.8949	2.8796	0.0010	0.4059
65.00	1.00	15.25	50.00	10.00	0.1119	0.0303	0.6791	0.1838	0.2726	0.7873
65.00	1.00	15.25	50.00	20.00	0.1119	0.0564	0.6791	0.3425	0.2221	0.7223
65.00	1.00	15.25	50.00	30.00	0.1119	0.0763	0.6791	0.4632	0.1620	0.6499
65.00	1.00	15.25	50.00	40.00	0.1119	0.0900	0.6791	0.5463	0.1087	0.5897
65.00	1.00	15.25	50.00	50.00	0.1119	0.0988	0.6791	0.5998	0.0689	0.5467
65.00	1.00	15.25	50.00	60.00	0.1119	0.1042	0.6791	0.6326	0.0420	0.5185
65.00	1.00	15.25	50.00	70.00	0.1119	0.1074	0.6791	0.6521	0.0249	0.5011
65.00	1.00	15.25	50.00	80.00	0.1119	0.1093	0.6791	0.6635	0.0145	0.4907
65.00	1.00	15.25	50.00	90.00	0.1119	0.1104	0.6791	0.6701	0.0084	0.4845
65.00	1.00	15.25	50.00	100.00	0.1119	0.1110	0.6791	0.6739	0.0048	0.4810



Fig. 1



Fig. 2